

Find Your Favorite Files Fast WordPerfect Magazine

The included FAVES.WPM macro displays a menu of your frequently used files, keyboard merges or macros on your screen. All you have to do is scroll to the item you want and press (Enter). If you select a file, the macro retrieves the file to your screen. If you select a keyboard merge, the macro runs the merge. If you choose a macro, the FAVES macro invokes the macro you've chosen.

Usage

To use the macro, press Macro (Alt-F10), type "faves" and press (Enter). The menu displays with your customized list (see sample list in figure below). Use the (Up Arrow) and (Down Arrow) keys to highlight the item you wish to use and press (Enter). You can also press the corresponding mnemonic letter (i.e., the bolded letter) to activate the item. Note that the items in the list are just placeholders. See "Customization" below to make the macro your own.

Find Faves Faster!

For even quicker access to your most used files, merges and macros, rename the macro to an Alt-letter macro. Do this by pressing List (F5), typing the full path to your macros directory — like c:\wp51\macros — and pressing (Enter). Highlight FAVES.WPM and press (3) Move/Rename. Delete FAVES.WPM at the New name: prompt and type "altx.wpm," where "x" is your chosen letter name. Press (Enter) and Exit (F7) to return to the document screen.

Customization

This macro requires extensive customization to work with your personal files. To do so, press (Ctrl-F10), type "faves" and press 2 Edit. You are now in the macro editor and will see the macro commands as listed below (and without the numbered lines on the left). *The highlighted lines are those you need to change.*

For example, I have a favorite document called LTRMOM that I want add as the first item in the list. To do so, I must change the macro in four places as :

1. First, I go to line 17 and change `{^}D{^}ocument` to `{^}L{^}trmom`.
2. I then go to line 27 and change `d~Doc~D~Doc` to `l~Ltrmom~D~Ltrmom`.
3. Next I go to line 43 and change `1~Doc` to `1~Ltrmom`.
4. Finally, I go to line 46 and change `{LABEL}Doc~{CALL}CheckDoc~{Retrieve}Doc{Enter}{QUIT}` to `{LABEL}Ltrmom~{CALL}CheckDoc~{Retrieve}Ltrmom{Enter}{QUIT}`.

Now that you're onto how to customize the macro, you can change it for your own favorite documents, merges or macros. *Just remember to make consistent changes to all four places for each item.*

★ ★ ★

The macro is outlined in red below as seen in the macro editor (which doesn't display the numbered lines that are in this document only for reference):

```

1 {DESCRIPTION}Document Management Macro~
2 {;}*****~
3 {;}As published in: WordPerfect Magazine~
4 {;}"Find Your Favorite Files Fast," May, 1994~
5 {;}(c) Copyright 1994 WordPerfect Corporation~
6 {;}*****~

7 {ON CANCEL}{GO}Cancel~~
8 {ASSIGN}ItemNum~7~
9 {FOR}R~1~{VARIABLE}ItemNum~~1~
10     {ASSIGN}{VARIABLE}R~R~~
11 {END FOR}
12 {ASSIGN}1R~{^R}~{ASSIGN}Number~1~

13 {LABEL}Top~
14 {DISPLAY ON}
15 {CHAR}Option~
16 {^P}{NTOK}28~{NTOK}8~6444444444444444444444447
17 {^P}{NTOK}28~{NTOK}9~5{VARIABLE}1R~    {^}]D{^\\}ocument    {^S}5
18 {^P}{NTOK}28~{NTOK}10~5{VARIABLE}2R~    {^}]L{^\\}etter    {^S}5
19 {^P}{NTOK}28~{NTOK}11~5{VARIABLE}3R~    {^}]M{^\\}emo    {^S}5
20 {^P}{NTOK}28~{NTOK}12~5{VARIABLE}4R~    {^}]I{^\\}nvoice    {^S}5
21 {^P}{NTOK}28~{NTOK}13~5{VARIABLE}5R~    {^}]F{^\\}axBox    {^S}5
22 {^P}{NTOK}28~{NTOK}14~5{VARIABLE}6R~    C{^}]a{^\\}lculator    {^S}5
23 {^P}{NTOK}28~{NTOK}15~5{VARIABLE}7R~    {^}]C{^\\}ancel    {^S}5
24 {^P}{NTOK}28~{NTOK}16~944444444444444444444448
25 {^P}{NTOK}1~{NTOK}25~~
26 {CASE}{VARIABLE}Option~~
27     d~Doc~D~Doc~I~Letter~L~Letter~m~Memo~M~Memo~
28     i~Invoice~I~Invoice~f~FaxBox~F~FaxBox~a~Calculator~
29     A~Calculator~c~Cancel~C~Cancel~
30     {Down}~Down~{Up}~Up~{Enter}~Enter~{OTHERWISE}~Beep~~

31 {LABEL}Down~
32 {IF}{VARIABLE}Number~={VARIABLE}ItemNum~~{GO}Top~{END IF}
33 {ASSIGN}{VARIABLE}Number~R~~
34 {ASSIGN}Number~{VARIABLE}Number~+1~
35 {ASSIGN}{VARIABLE}Number~R~{^R}~{GO}Top~

```

```

36 {LABEL}Up~
37 {IF}{VARIABLE}Number~=1~{GO}Top~{END IF}
38 {ASSIGN}{VARIABLE}Number~R~~
39 {ASSIGN}Number~{VARIABLE}Number~-1~
40 {ASSIGN}{VARIABLE}Number~R~{^R}~{GO}Top~

41 {LABEL}Enter~
42 {CASE}{VARIABLE}Number~~
43     1~Doc~2~Letter~3~Memo~4~Invoice~5~FaxBox~6~Calculator~
44     7~Cancel~{OTHERWISE}~Top~~

45 {LABEL}Beep~{BELL}{GO}Top~
46 {LABEL}Doc~{CALL}CheckDoc~{Retrieve}Doc{Enter}{QUIT}
47 {LABEL}Letter~{CALL}CheckDoc~{Merge/Sort}1Letter{Enter}{Enter}{QUIT}
48 {LABEL}Memo~{CALL}CheckDoc~{Merge/Sort}1Memo{Enter}{Enter}{QUIT}
49 {LABEL}Invoice~{CALL}CheckDoc~{Merge/Sort}1Invoice{Enter}{Enter}{QUIT}
50 {LABEL}FaxBox~{Screen}{Screen}{NEST}FaxBox~{QUIT}
51 {LABEL}Calculator~{Screen}{Screen}{NEST}Calc~{QUIT}
52 {LABEL}Cancel~{Screen}{Screen}{QUIT}

53 {LABEL}CheckDoc~
54 {IF}{SYSTEM}Document~!=256~
55     {BELL}{PROMPT}{^R}Clear Screen, Rerun Macro{^S}~
56     {WAIT}30~{QUIT}
57 {END IF}
58 {RETURN}

```

